Pensieve header: The double and meta-double of the 2D pencil; continues pensieve://2017-05/.

Issues:

1. Improve DUForm.

```
SetDirectory["C:\\drorbn\\AcademicPensieve\\2017-06"];
```

The “degree carrier” is $\hbar$, and all “coupling constants” are proportional to it.

```
$TD = 3; ℏ /: ℏ^d_· /; d > $TD := 0;
```

The 2D Lie BiAlgebra Pencil

We hope to stick to $A = e^{\hbar\beta a}$ and to $B = e^{\hbar\alpha b}$, where $[a, x] = \alpha x$ and $[b, y] = -\beta y$.
Also, $\Delta_{12}(a, A, x, b, B, y) = (a_1 + a_2,\ A_1 A_2,\ x_1 + A_1 x_2,\ b_1 + b_2,\ B_1 B_2,\ y_1 B_2 + y_2)$.
Also, (a, x) and $\hbar(b, y)$ are dual bases.

```
AlgebraAtom = a | A[_] | x | b | B[_] | y;
$PBWRule = {A[_] → 1, a → 2, x → 3, y → 4, B[_] → 5, b → 6};
```

```
B[a, x] = α x; B[x, A[n_]] = (e^(n ℏ α β) - 1) U[A[n], x]; B[a, A[_]] = 0;
B[y, b] = β y; B[B[n_], y] = (e^(n ℏ α β) - 1) U[y, B[n]]; B[b, B[_]] = 0;
Δ[a] = U1[a] U2[] + U1[] U2[a]; Δ[A[n_]] := U1[A[n]] U2[A[n]];
Δ[x] = U1[x] U2[] + U1[A[1]] U2[x];
Δ[b] = U1[b] U2[] + U1[] U2[b]; Δ[B[n_]] := U1[B[n]] U2[B[n]];
Δ[y] = U1[y] U2[B[1]] + U1[] U2[y];
S[a] = -a; S[A[n_]] := A[-n]; S[x] = - U[A[-1], x];
S[b] = -b; S[B[n_]] := B[-n]; S[y] = - U[y, B[-1]];
Si[a] = -a; Si[A[n_]] := A[-n]; Si[x] = - U[x, A[-1]];
Si[b] = -b; Si[B[n_]] := B[-n]; Si[y] = - U[B[-1], y];
(* This extra line is annoying *)
```

```
ExpandAB[ℰ_] := Expand@Normal@Series[ℰ //. {
     c_. U_i_[λ___, A[n_], ρ___] :>
       Expand[c Sum[((-1)^d ℏ^d β^d n^d)/d! U_i[λ, Sequence @@ Table[a, {d}], ρ], {d, 0, $TD}]],
     c_. U_i_[λ___, B[n_], ρ___] :> Expand[
        c Sum[((-1)^d ℏ^d α^d n^d)/d! U_i[λ, Sequence @@ Table[b, {d}], ρ], {d, 0, $TD}]]
    },
   {ℏ,
    0,
    $TD}]
```

UEA with provisional modification

This section is based on penseive://Projects/UEA/.

```
B[0, _] = 0; B[_, 0] = 0;
B[c_ * x : AlgebraAtom, y_] := Expand[c B[x, y]];
B[y_, c_ * x : AlgebraAtom] := Expand[c B[y, x]];
B[x_Plus, y_] := B[#, y] & /@ x;
B[x_, y_Plus] := B[x, #] & /@ y;
B[x_, x_] = 0;
B[y_, x_] := Expand[-B[x, y]];
```

```
x_ ≤ y_ := OrderedQ[{x, y} /. $PBWRule]; x_ ≺ y_ := ! OrderedQ[{y, x} /. $PBWRule];
UU_i_[] := U_i[]; UU_i_[1] := U_i[];
UU_i_[x_[n_]^p_] := U_i[x[n p]];
UU_i_[x_^p_] := UU_i @@ Table[x, {p}];
UU_i_[ℰ_] := ℰ /. {
    U[xs__] :→ UU_i[xs],
    x : AlgebraAtom :→ U_i[x]
   };
UU_i_[x_, xs__] := UU_t1[x] UU_t2[xs] // Expand // m_{t1,t2→i};
USimp[ℰ_] := Collect[ℰ, Times[U_[__] ..], Expand];
USimp[ℰ_] := Expand[ℰ];
```

```
m_s__[0] = 0;
m_s__[x_Plus] := m_s /@ x;
m_{i_→j_}[ℰ_] := ℰ /. U_i → U_j;
```

```
m_{i_,j_→k_}[c_. U_i_[x___] U_j_[]] := c U_k[x];
m_{i_,j_→k_}[c_. U_i_[] U_j_[y___]] := c U_k[y];
m_{i_,j_→k_}[c_. U_i_[xx___, x_[n1_]] U_j_[x_[n2_], yy___]] :=
  USimp[c If[TrueQ[n1 + n2 == 0], U_i[xx] U_j[yy], U_i[xx, x[n1 + n2]] U_j[yy]] // m_{i,j→k}];
m_{i_,j_→k_}[c_. U_i_[xx___, x_] U_j_[y_, yy___]] := If[x ≤ y,
   c U_k[xx, x, y, yy],
   ((U_i[xx] (U_j[y, x] + UU_j[B[x, y]]) // Expand // m_{i,j→i}) U_j[yy] // Expand // m_{i,j→k})
     c // USimp
  ];
```

```
Supp[ℰ_] := Union@Cases[{ℰ}, U_i_[___] :→ i, ∞];
```

```
Unprotect[NonCommutativeMultiply];
NonCommutativeMultiply[x_] := x;
x_ ** y_ := Module[
   {Sx = Supp[x], Sy = Supp[y], is, σ, z},
   If[MatchQ[Sx ⋃ Sy, {_Integer ...}] && Min[Sx ⋃ Sy] < 0,
    is = Abs[Sx] ⋂ Abs[Sy];
    z = x; Do[z = m_{-i→-σ@i}[m_{i→σ@i}[z]], {i, is}];
    z = USimp[y z]; Do[z = dm_{σ@i,i→i}[z], {i, is}];
    z,
    (* else *) is = Sx ⋂ Sy;
    z = x; Do[z = m_{i→σ@i}[z], {i, is}];
    z = USimp[y z]; Do[z = m_{σ@i,i→i}[z], {i, is}];
    z
   ]
  ];
UB[x_, y_] := USimp[x ** y - y ** x];
```

```
𝕆[func_] := Normal@Series[func, {ℏ, 0, $TD}];
𝕆[specs_, func_] := Module[{rules, vars, elems},
  rules = Union @@ Cases[{specs}, U_[u___] :> Cases[{u}, r_Rule], ∞];
  vars = First /@ rules; elems = Last /@ rules;
  USimp@Total[CoefficientRules[𝕆[func], vars] /. (ps_ → c_) :> c (
       specs /. MapThread[((#1 → _) :> #3^#2) &, {vars, ps, elems}] /. U_i_ :> UU_i
      )]
 ]
```

The 2D Lie BiAlgebra Pencil, Testing

```
𝕆[U_1[a → a], ⅇ^(ℏ β a)]
```

$U_1[\,] + \beta\,\hbar\,U_1[a] + \frac{1}{2}\beta^2\hbar^2\,U_1[a, a] + \frac{1}{6}\beta^3\hbar^3\,U_1[a, a, a]$

```
USimp@With[{An = 𝕆[U_1[a → a], ⅇ^(-n ℏ β a)]}, UB[U_1[x], An] - 𝕆[ⅇ^(n ℏ α β) - 1] An ** U_1[x]]
```

0

```
B[x, A[3]]
```

$\left(-1 + e^{3\alpha\beta\hbar}\right) U[A[3], x]$

```
$TD = 6;
USimp@With[{Bn = 𝕆[U_1[b → b], ⅇ^(-n ℏ α b)]}, UB[Bn, U_1[y]] - 𝕆[ⅇ^(n ℏ α β) - 1] U_1[y] ** Bn]
```

0

```
z = U_1[a, A[2], x, x, x] U_2[a, a, x] U_3[a, a, A[-3], x];
(z // m_{1,2→1} // m_{1,3→1}) - (z // m_{2,3→2} // m_{1,2→1})
```

0

```
z = U1[y, y, y, b, B[2]] U2[y, b, b] U3[y, b, b, B[-3]];
(z // m1,2→1 // m1,3→1) - (z // m2,3→2 // m1,2→1)
```

```
0
```

The Co-Product and Co-Associativity

```
Δi_→j_,k_[ℰ_] := USimp@Module[{tj, tk}, ℰ /. {
      Ui[] → Uj[] Uk[],
      Ui[f_, fs___] :→
       (USimp[(Δ[f] /. {U1 → Uj, U2 → Uk}) Δi→tj,tk[Ui[fs]]] // mj,tj→j // mk,tk→k)
     }];
Δi_→j_,k_,l__[ℰ_] := ℰ // Δi→j,k // Δk→k,l
```

```
Δ1→1,2[U1[#]] & /@ {a, A[7], x, y, b, B[-3]}
```

```
{U1[a] U2[] + U1[] U2[a], U1[A[7]] U2[A[7]], U1[x] U2[] + U1[A[1]] U2[x],
 U1[] U2[y] + U1[y] U2[B[1]], U1[b] U2[] + U1[] U2[b], U1[B[-3]] U2[B[-3]]}
```

```
{lhs = U1[x] // Δ1→1,2 // Δ2→2,3, rhs = U1[x] // Δ1→1,3 // Δ1→1,2, lhs == rhs}
```

```
{U1[x] U2[] U3[] + U1[A[1]] U2[x] U3[] + U1[A[1]] U2[A[1]] U3[x],
 U1[x] U2[] U3[] + U1[A[1]] U2[x] U3[] + U1[A[1]] U2[A[1]] U3[x], True}
```

```
U1[y] // Δ1→1,2
```

```
U1[] U2[y] + U1[y] U2[B[1]]
```

```
{lhs = U1[y] // Δ1→1,2 // Δ2→2,3, rhs = U1[y] // Δ1→1,3 // Δ1→1,2, lhs == rhs}
```

```
{U1[] U2[] U3[y] + U1[] U2[y] U3[B[1]] + U1[y] U2[B[1]] U3[B[1]],
 U1[] U2[] U3[y] + U1[] U2[y] U3[B[1]] + U1[y] U2[B[1]] U3[B[1]], True}
```

```
z = U1[a, A[2], x, x, x] U2[a, a, A[-3], x];
(z // m1,2→1 // Δ1→1,2) - (z // Δ2→3,4 // Δ1→1,2 // m1,3→1 // m2,4→2)
```

```
0
```

```
z = U1[y, y, y, b, B[2]] U2[y, b, b, B[-3]];
(z // m1,2→1 // Δ1→1,2) - (z // Δ2→3,4 // Δ1→1,2 // m1,3→1 // m2,4→2)
```

```
0
```

The Antipode

```
Si_[ℰ_] := Module[{ti}, USimp[
    ℰ /. Ui[x_, xs___] :→ mti,i→i[Expand[UUi[S[x]] Sti[Uti[xs]]]]
   ]];
Sii_[ℰ_] := Module[{ti}, USimp[
    ℰ /. Ui[x_, xs___] :→ mti,i→i[Expand[UUi[Si[x]] Siti[Uti[xs]]]]
   ]];
```

```
{U1[x] // S1 // S1, U1[y] // S1 // S1}
```

$\{e^{\alpha\beta\hbar} U_1[x], e^{\alpha\beta\hbar} U_1[y]\}$

```
{U1[x] // S1 // Si1, U1[y] // S1 // Si1}
```

$\{U_1[x], U_1[y]\}$

```
{z = U1[]; (z // Δ1→1,2 // S1 // m1,2→1), z = U1[]; (z // Δ1→1,2 // S2 // m1,2→1)}
```

$\{U_1[], U_1[]\}$

```
z = U1[a, A[3], x, x];
{z // Δ1→1,2 // S2 // m1,2→1, z // Δ1→1,2 // S1 // m1,2→1}
```

{0, 0}

```
z = U1[y, y, y, b, B[-3]];
{z // Δ1→1,2 // S2 // m1,2→1, z // Δ1→1,2 // S1 // m1,2→1}
```

{0, 0}

```
{z = U1[a, A[2], x, x, x] U2[a, a, A[-3], x]; (z // m1,2→1 // S1) - (z // S1 // S2 // m2,1→1),
 z = U1[y, y, y, b, B[2]] U2[y, b, b, b, B[6]]; (z // m1,2→1 // S1) - (z // S1 // S2 // m2,1→1)}
```

{0, 0}

```
{z = U1[a, A[2], x, x, x]; (z // S1 // Δ1→2,1) - (z // Δ1→1,2 // S1 // S2),
 z = U1[y, y, y, b, B[2]]; (z // S1 // Δ1→2,1) - (z // Δ1→1,2 // S1 // S2)
}
```

{0, 0}

The Pairing

```
P[U[], U[B[_]]] = P[U[A[_]], U[]] = 1;
P[U[], U[__]] = P[U[__], U[]] = 0;
(   P[U[a], U[b]] = ℏ^-1          P[U[a], U[B[n_]]] = -n α          P[U[a], U[y]] = 0
 P[U[A[n_]], U[b]] = -n β   P[U[A[n_]], U[B[m_]]] = e^(n m ℏ α β)   P[U[A[_]], U[y]] = 0
    P[U[x], U[b]] = 0              P[U[x], U[B[_]]] = 0             P[U[x], U[y]] = ℏ^-1  );
```

```
P[U[], U[]] = 1;
P_{i_,j_}[ℰ_] := USimp[ℰ /. U_i[xs___] U_j[ys___] → P[U[xs], U[ys]]];
```

The pairing sequence: ⟨one,one⟩ (above), ⟨many,one⟩, ⟨many,many⟩.

```
P[U[x_, xs__], U[y_]] := P[U[x, xs], U[y]] =
   Module[{i, j, k, l}, USimp[U_i[x] UU_j[xs] Δ_{k→k,l}[U_k[y]]] // P_{i,k} // P_{j,l}];
P[U[xs__], U[y_, ys__]] := P[U[xs], U[y, ys]] =
   Module[{i, j, k, l}, USimp[Δ_{i→i,j}[UU_i[xs]] U_k[y] UU_l[ys]] // P_{i,k} // P_{j,l}];
```

```
z = U_i[a] U_j[x] U_k[y];
{m_{i,j→i}[z] - m_{j,i→i}[z], Δ_{k→k,l}[z] - Δ_{k→l,k}[z]}
```

$$\{\alpha\, U_i[x]\, U_k[y],\ -U_i[a]\, U_j[x]\, U_k[y]\, U_l[] + U_i[a]\, U_j[x]\, U_k[]\, U_l[y] - U_i[a]\, U_j[x]\, U_k[B[1]]\, U_l[y] + U_i[a]\, U_j[x]\, U_k[y]\, U_l[B[1]]\}$$

```
Table[z = U_i[xi] U_j[xj] U_k[yk];
 {(m_{i,j→i}[z] - m_{j,i→i}[z]) // P_{i,k}, (Δ_{k→k,l}[z] - Δ_{k→l,k}[z]) // P_{i,k} // P_{j,l}},
 {xi, {a, x}}, {xj, {a, x}}, {yk, {b, y}}]
```

$$\left\{\left\{\{\{0,0\},\{0,0\}\},\left\{\{0,0\},\left\{\frac{\alpha}{\hbar},\frac{\alpha}{\hbar}\right\}\right\}\right\},\left\{\left\{\{0,0\},\left\{-\frac{\alpha}{\hbar},-\frac{\alpha}{\hbar}\right\}\right\},\{\{0,0\},\{0,0\}\}\right\}\right\}$$

```
Table[z = U_i[xi] U_k[yk] U_l[yl];
 {(Δ_{i→i,j}[z] - Δ_{i→j,i}[z]) // P_{i,k} // P_{j,l}, (m_{k,l→k}[z] - m_{l,k→k}[z]) // P_{i,k}},
 {xi, {a, x}}, {yk, {b, y}}, {yl, {b, y}}]
```

$$\left\{\{\{\{0,0\},\{0,0\}\},\{\{0,0\},\{0,0\}\}\},\left\{\left\{\{0,0\},\left\{-\frac{\beta}{\hbar},-\frac{\beta}{\hbar}\right\}\right\},\left\{\left\{\frac{\beta}{\hbar},\frac{\beta}{\hbar}\right\},\{0,0\}\right\}\right\}\right\}$$

```
lhs = Factor@Table[ℏ^n P[U @@ Table[x, {n}], U @@ Table[y, {n}]], {n, $TD = 7}]
```

$$\Big\{1,\ 1+e^{\alpha\beta\hbar},\ (1+e^{\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}),\ (1+e^{\alpha\beta\hbar})^2(1+e^{2\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}),$$
$$(1+e^{\alpha\beta\hbar})^2(1+e^{2\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}+e^{3\alpha\beta\hbar}+e^{4\alpha\beta\hbar}),$$
$$(1+e^{\alpha\beta\hbar})^3(1+e^{2\alpha\beta\hbar})(1-e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar})^2(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}+e^{3\alpha\beta\hbar}+e^{4\alpha\beta\hbar}),$$
$$(1+e^{\alpha\beta\hbar})^3(1+e^{2\alpha\beta\hbar})(1-e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar})^2$$
$$(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}+e^{3\alpha\beta\hbar}+e^{4\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}+e^{3\alpha\beta\hbar}+e^{4\alpha\beta\hbar}+e^{5\alpha\beta\hbar}+e^{6\alpha\beta\hbar})\Big\}$$

```
rhs = Simplify@FunctionExpand@Table[QFactorial[n, e^{ℏ α β}], {n, $TD = 7}]
```

$$\Big\{1,\ 1+e^{\alpha\beta\hbar},\ (1+e^{\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}),\ (1+e^{\alpha\beta\hbar})^2(1+e^{2\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}),$$
$$(1+e^{\alpha\beta\hbar})^2(1+e^{2\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}+e^{3\alpha\beta\hbar}+e^{4\alpha\beta\hbar}),$$
$$(1+e^{\alpha\beta\hbar})^3(1+e^{2\alpha\beta\hbar})(1-e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar})^2(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}+e^{3\alpha\beta\hbar}+e^{4\alpha\beta\hbar}),$$
$$(1+e^{\alpha\beta\hbar})^3(1+e^{2\alpha\beta\hbar})(1-e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar})^2$$
$$(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}+e^{3\alpha\beta\hbar}+e^{4\alpha\beta\hbar})(1+e^{\alpha\beta\hbar}+e^{2\alpha\beta\hbar}+e^{3\alpha\beta\hbar}+e^{4\alpha\beta\hbar}+e^{5\alpha\beta\hbar}+e^{6\alpha\beta\hbar})\Big\}$$

```
MapThread[Equal, {lhs, rhs}]
```

{True, True, True, True, True, True, True}

```
P[U[a, a, a, a, a], U[b, b, b, b, b]]
```

$$\frac{120}{\hbar^5}$$

```
Table[P[z1, z2],
  {z1, {U[], U[a], U[x], U[a, a], U[a, x], U[x, x],
    U[a, a, a], U[a, a, x], U[a, x, x], U[x, x, x]}}, {z2, {U[], U[b], U[y],
    U[b, b], U[y, b], U[y, y], U[b, b, b], U[y, b, b], U[y, y, b], U[y, y, y]}}
 ] //
 MatrixForm
```

$$\begin{pmatrix}
1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & \frac{1}{\hbar} & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & \frac{1}{\hbar} & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & \frac{2}{\hbar^2} & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & \frac{1}{\hbar^2} & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & \frac{1}{\hbar^2}+\frac{e^{\alpha\beta\hbar}}{\hbar^2} & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & \frac{6}{\hbar^3} & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{2}{\hbar^3} & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{1}{\hbar^3}+\frac{e^{\alpha\beta\hbar}}{\hbar^3} & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{1}{\hbar^3}+\frac{2e^{\alpha\beta\hbar}}{\hbar^3}+\frac{2e^{2\alpha\beta\hbar}}{\hbar^3}+\frac{e^{3\alpha\beta\hbar}}{\hbar^3}
\end{pmatrix}$$

Pairing axioms:

```
$TD = 5;
(U₁[a, x, x] S₋₁[U₋₁[y, y, b]] // Expand // P₁,₋₁) -
  (S₁[U₁[a, x, x]] U₋₁[y, y, b] // Expand // P₁,₋₁) // ExpandAB
```

0

```
((U₁[a, x, x] U₂[a, A[1], x] U₋₁[y, y, y, b, b, b] // m₁,₂→₁ // Expand) // P₁,₋₁) -
 ((U₁[a, x, x] U₂[a, A[1], x] U₋₁[y, y, y, b, b, b] // Δ₋₁->₋₁,₋₂ // Expand) // P₁,₋₁ // P₂,₋₂)
```

0

```
((U₁[a, x, x, x, x] U₋₂[y, B[1], b] U₋₁[y, y, y, b, b, b] // m₋₁,₋₂→₋₁ // Expand) // P₁,₋₁) -
 ((U₁[a, x, x, x, x] U₋₂[y, B[1], b] U₋₁[y, y, y, b, b, b] // Δ₁->₁,₂ // Expand) // P₁,₋₁ // P₂,₋₂)
```

0

The Double

```
DUForm[ℰ_] := ℰ //. Uᵢ_[fs___] Uⱼ_[gs___] /; i + j == 0 ∧ j > 0 ⧴ "DU"ⱼ[fs, gs];
DUᵢ_[a] := U₋ᵢ[] Uᵢ[a]; DUᵢ_[A[n_]] := U₋ᵢ[] Uᵢ[A[n]]; DUᵢ_[x] := U₋ᵢ[] Uᵢ[x];
DUᵢ_[y] := U₋ᵢ[y] Uᵢ[]; DUᵢ_[b] := U₋ᵢ[b] Uᵢ[]; DUᵢ_[B[n_]] := U₋ᵢ[B[n]] Uᵢ[];
DUᵢ_[] := U₋ᵢ[] Uᵢ[];
```

```
DU₁ /@ {a, A[1], x, y, b, B[-1]}
```

{U₋₁[] U₁[a], U₋₁[] U₁[A[1]], U₋₁[] U₁[x], U₋₁[y] U₁[], U₋₁[b] U₁[], U₋₁[B[-1]] U₁[]}

```
dm_{i_,j_→k_}[ℰ_] := Module[{t1, t2, t3, h1, h2, h3},
   ℰ // Δ_{-j→t1,t2,t3} // S_{t3} // Δ_{i→h1,h2,h3} // P_{h1,t1} // P_{h3,t3} // m_{j,h2→k} // m_{-i,t2→-k}];
dΔ_{i_→j_,k_}[ℰ_] := ℰ // Δ_{i→j,k} // Δ_{-i→-j,-k};
dS_{i_}[ℰ_] := Module[{h}, (ℰ // m_{i→h} // Si_h // S_{-i}) U_{-h}[] U_i[] // Expand // dm_{h,i→i}];
```

```
Module[{bas},
  bas = Join[{DU_1[]}, DU_1 /@ {a, A[n], x, y, B[m], b}, {β DU_1[a] + α DU_1[b]}];
  Table[If[f ** g =!= g ** f ∨ f === DU_1[] ∨ g === DU_1[], f ** g, "="], {f, bas}, {g, bas}] //
   ReplacePart[{1, 1} → "**"]
 ] // DUForm // MatrixForm
```

$**$	$DU_1[a]$	$DU_1[A[n]]$	$DU_1[x]$	$DU_1[y]$
$DU_1[a]$	=	=	$-\alpha\, DU_1[x] + DU_1[a, x]$	$\alpha\, DU_1[y] + DU_1[y, a]$
$DU_1[A[n]]$	=	=	$e^{n\alpha\beta\hbar} DU_1[A[n], x]$	$e^{-n\alpha\beta\hbar} DU_1[y, A[n]]$
$DU_1[x]$	$DU_1[a, x]$	$DU_1[A[n], x]$	=	$-\frac{DU_1[A[1]]}{\hbar} + \frac{DU_1[B[1]]}{\hbar} + DU_1[y, x]$
$DU_1[y]$	$DU_1[y, a]$	$DU_1[y, A[n]]$	$DU_1[y, x]$	=
$DU_1[B[m]]$	=	=	$DU_1[B[m], x]$	$e^{m\alpha\beta\hbar} DU_1[y, B[m]]$
$DU_1[b]$	=	=	$DU_1[b, x]$	$-\beta\, DU_1[y] + DU_1[y, b]$
$\beta\, DU_1[a] + \alpha\, DU_1[b]$	=	=	=	=

```
{DU_1[A[1]] ** DU_1[x], DU_1[A[1]] ** DU_1[y],
  DU_1[x] ** DU_1[B[1]], DU_1[B[1]] ** DU_1[y]} // DUForm
```

$\{e^{\alpha\beta\hbar} DU_1[A[1], x],\ e^{-\alpha\beta\hbar} DU_1[y, A[1]],\ e^{\alpha\beta\hbar} DU_1[B[1], x],\ e^{\alpha\beta\hbar} DU_1[y, B[1]]\}$

```
DU_1[x] DU_2[a] // dm_{1,2→1} // DUForm
```

$DU_1[a, x]$

```
U_{-1}[] U_1[a] U_{-2}[] U_2[x] // dm_{1,2→1} // DUForm
```

$-\alpha\, DU_1[x] + DU_1[a, x]$

```
U_{-1}[] U_1[a] U_{-2}[b] U_2[] // dm_{1,2→1} // DUForm
```

$DU_1[b, a]$

```
{U_{-1}[] U_1[a] U_{-2}[y] U_2[] // dm_{2,1→1}, U_{-1}[] U_1[a] U_{-2}[y] U_2[] // dm_{1,2→1}} // DUForm
```

$\{DU_1[y, a],\ \alpha\, DU_1[y] + DU_1[y, a]\}$

```
(U_{-1}[] U_1[x] U_{-2}[b] U_2[] - U_{-1}[b] U_1[] U_{-2}[] U_2[x]) // dm_{1,2→1} // DUForm
```

$-\beta\, DU_1[x]$

```
U_{-1}[] U_1[A[1]] U_{-2}[y] U_2[] // dm_{1,2→1}
```

$e^{-\alpha\beta\hbar} U_{-1}[y]\, U_1[A[1]]$

```
U_{-1}[] U_1[x] U_{-2}[b] U_2[] // dm_{1,2→1}
```

$-\beta\, U_{-1}[]\, U_1[x] + U_{-1}[b]\, U_1[x]$

```
U_{-1}[] U_1[x] U_{-2}[B[1]] U_2[] // dm_{1,2→1}
```

$e^{\alpha\beta\hbar} U_{-1}[B[1]]\, U_1[x]$

```
DU₁[x] DU₂[y] // dm1,2→1 // DUForm
```

$$-\frac{\mathrm{DU}_1[\mathrm{A}[1]]}{\hbar}+\frac{\mathrm{DU}_1[\mathrm{B}[1]]}{\hbar}+\mathrm{DU}_1[y, x]$$

```
z = U₋₁[] U₁[x] U₋₂[y] U₂[] U₋₃[b] U₃[];
(z // dm1,2→1 // dm1,3→1) - (z // dm2,3→2 // dm1,2→1) // DUForm
```

0

```
$TD = 5;
z = U₋₁[y, b, b] U₁[A[2], a, x, x] U₋₂[B[-1], y, y, b, b] U₂[a] U₋₃[y, y, b] U₃[a, a, x, x, x];
(z // dm1,2→1 // dm1,3→1) - (z // dm2,3→2 // dm1,2→1) // DUForm
```

0

```
(# → DUForm[dS₁[DU₁[#]]]) & /@ {a, x, y, b}
```

$\{a \to -\mathrm{DU}_1[a],\ x \to -e^{-\alpha\beta\hbar}\,\mathrm{DU}_1[\mathrm{A}[-1], x],\ y \to -\mathrm{DU}_1[y, \mathrm{B}[-1]],\ b \to -\mathrm{DU}_1[b]\}$

```
(DU₁[x] DU₂[y] // dm1,2→1 // dS₁) - (DU₁[x] DU₂[y] // dS₁ // dS₂ // dm2,1→1) // ExpandAB
```

0

Coassociativity

```
(lhs = DU₁[#] // dΔ1→1,2 // dΔ2→2,3; rhs = DU₁[#] // dΔ1→1,3 // dΔ1→1,2; lhs == rhs) & /@
 {a, x, y, b}
```

{True, True, True, True}

dΔ is algebra morphism

```
z = U₋₁[y, y, y, b, B[2]] U₁[x] U₋₂[y, b, b, B[-3]] U₂[a, x];
(z // dm1,2→1 // dΔ1→1,2) - (z // dΔ2→3,4 // dΔ1→1,2 // dm1,3→1 // dm2,4→2)
```

0

dS is algebra anti-morphism

dS is the convolution inverse of dm.

```
z = U₋₁[y, y, y, b, B[-3]] U₁[a, x, x];
{z // dΔ1→1,2 // dS₂ // dm1,2→1, z // dΔ1→1,2 // dS₁ // dm1,2→1}
```

{0, 0}

The R-Matrix

Quesne's formulas:

$$q = e^{\hbar\alpha\beta};\ \ e_{q\_}[x\_] := \mathrm{Exp}\left[\sum_{k=1}^{\$TD}\frac{(1-q)^k}{k\,(1-q^k)}\,x^k\right];$$

```
Table[Together@SeriesCoefficient[𝕖ρ[x], {x, 0, n}], {n, 0, $TD}]
```

$$\left\{1, 1, \frac{1}{1+\rho}, \frac{1}{(1+\rho)(1+\rho+\rho^2)}, \frac{1}{(1+\rho)^2(1+\rho^2)(1+\rho+\rho^2)}, 1\Big/\left((1+\rho)^2(1+\rho^2)(1+\rho+\rho^2)(1+\rho+\rho^2+\rho^3+\rho^4)\right)\right\}$$

```
R_{i_,j_} := 𝕆[U_{-i}[y → y, b → b] U_i[] U_{-j}[] U_j[a → a, x → x], 𝕖^{ℏ b a} 𝕖_q[ℏ x y]]
```

```
$TD = 3; R_{1,2} // DUForm
```

$DU_1[\,]\, DU_2[\,] + \hbar\, DU_1[b]\, DU_2[a] + \hbar\, DU_1[y]\, DU_2[x] + \frac{1}{2}\hbar^2 DU_1[b, b]\, DU_2[a, a] + \hbar^2 DU_1[y, b]\, DU_2[a, x] + \frac{1}{2}\hbar^2 DU_1[y, y]\, DU_2[x, x] - \frac{1}{4}\alpha\beta\hbar^3 DU_1[y, y]\, DU_2[x, x] + \frac{1}{6}\hbar^3 DU_1[b, b, b]\, DU_2[a, a, a] + \frac{1}{2}\hbar^3 DU_1[y, b, b]\, DU_2[a, a, x] + \frac{1}{2}\hbar^3 DU_1[y, y, b]\, DU_2[a, x, x] + \frac{1}{6}\hbar^3 DU_1[y, y, y]\, DU_2[x, x, x]$

```
$TD = 5; {lhs = R_{1,3} // dΔ_{1→1,2}, rhs = R_{2,3} ** R_{1,3}, lhs - rhs} // Last // ExpandAB // DUForm
```

0

```
$TD = 5; {lhs = R_{1,2} // dΔ_{2→2,3}, rhs = R_{1,2} ** R_{1,3}, lhs - rhs} // Last // ExpandAB // DUForm
```

0

```
$TD = 5; Table[
 {lhs = R_{1,2} ** dΔ_{1→2,1}[f], rhs = dΔ_{1→1,2}[f] ** R_{1,2}, lhs - rhs} // Last // ExpandAB // DUForm,
 {f, {U_{-1}[] U_1[a], U_{-1}[] U_1[x], U_{-1}[b] U_1[], U_{-1}[y] U_1[]}}]
```

{0, 0, 0, 0}

```
$TD = 5;
{lhs = ExpandAB[R_{1,2} ** R_{1,3} ** R_{2,3}],
  rhs = ExpandAB[R_{2,3} ** R_{1,3} ** R_{1,2}], Coefficient[lhs - rhs, ℏ^$TD]} // Last
```

0

```
$TD = 5; Si_2[R_{1,2}] ** R_{1,2} // ExpandAB // DUForm
```

$DU_1[\,]\, DU_2[\,]$

Cuaps (Unfinished)

Drinfeld element *u*.

```
u_{i_} := R_{1,2} // dS_1 // dm_{2,1→i}
```

```
$TD = 2; u1 // ExpandAB // DUForm
```

$\text{DU}_1[\,] - \beta\hbar\,\text{DU}_1[\text{a}] + \alpha\hbar\,\text{DU}_1[\text{b}] + \frac{1}{2}\beta^2\hbar^2\,\text{DU}_1[\text{a, a}] - \hbar\,\text{DU}_1[\text{b, a}] - \alpha\beta\hbar^2\,\text{DU}_1[\text{b, a}] + \frac{1}{2}\alpha^2\hbar^2\,\text{DU}_1[\text{b, b}] - \hbar\,\text{DU}_1[\text{y, x}] + \alpha\beta\hbar^2\,\text{DU}_1[\text{y, x}] + \beta\hbar^2\,\text{DU}_1[\text{b, a, a}] - \alpha\hbar^2\,\text{DU}_1[\text{b, b, a}] - 2\alpha\hbar^2\,\text{DU}_1[\text{y, b, x}] + \frac{1}{2}\hbar^2\,\text{DU}_1[\text{b, b, a, a}] + \hbar^2\,\text{DU}_1[\text{y, b, a, x}] + \frac{1}{2}\hbar^2\,\text{DU}_1[\text{y, y, x, x}]$

Conjugation by the Drinfeld element implements the square of the antipode.

```
$TD = 3; Table[DU1[f] ** u1 == u1 ** dS1[dS1[DU1[f]]] // ExpandAB, {f, {a, x, b, y}}]
```

{True, True, True, True}

Inverse of the Drinfeld element

```
ui_i_ := R_1,2 // dS2 // dS2 // dm_2,1→i
```

```
$TD = 3; ui1 ** u1 // ExpandAB
```

$U_{-1}[\,]\ U_1[\,]$

Drinfeld commutes with its antipode and the product is central

```
$TD = 3; u1 ** dS1[u1] - dS1[u1] ** u1 // ExpandAB
```

0

```
$TD = 3;
(u1 ** dS1[u1] ** DU1[#] - DU1[#] ** u1 ** dS1[u1]) & /@ {a, x, y, b} // ExpandAB
```

{0, 0, 0, 0}

Multiplying the inverse and antipode of Drinfeld

```
$TD = 3; ui1 ** dS1[u1] - U_-1[B[1]] U1[A[-1]] // ExpandAB // DUForm
```

0

This implies that dS[u] = $u\,\text{BA}^{-1}$. In other words, u dS[u] = $u^2\,\text{BA}^{-1}$ so we can take the square root and call it v. This is the ribbon element.

```
v_i_ := (𝕆[U_-i[b → b] U_i[a → a], Exp[ħ/2 (-α b + β a)]]) ** u_i
```

```
$TD = 3; u1 ** dS1[u1] - v1 ** v1 // ExpandAB
```

0

v is supposed to be the Reidemeister 1 curl (with the appropriate cuaps/spinners added!). Note how it too is central.

```
$TD = 3; (v1 ** DU1[#] - DU1[#] ** v1) & /@ {a, x, y, b} // ExpandAB
```

{0, 0, 0, 0}

The rule for the cuaps is to add to a right-moving cup uv^{-1} and to a right – moving cap vu^{-1}. Note how

$vu^{-1} = B^{1/2} A^{-1/2}$

```
$TD = 3; v₁ ** ui₁ - 𝕆[U₋₁[b → b] U₁[a → a], Exp[ħ/2 (-α b + β a)]] // ExpandAB
```

0

```
cap_i_ := 𝕆[U₋ᵢ[b → b] Uᵢ[a → a], Exp[ħ/2 (-α b + β a)]]
cup_i_ := 𝕆[U₋ᵢ[b → b] Uᵢ[a → a], Exp[-ħ/2 (-α b + β a)]]
```

Now let's do R2 and oppositely oriented Reidemeister 2:

```
$TD = 3; dS₂[R₁,₂] R₃,₄ // dm₁,₃→₁ // dm₂,₄→₂ // ExpandAB // DUForm
```

DU₁[] DU₂[]

```
$TD = 3;
dS₂[R₁,₂] R₃,₄ cup₅ cap₆ // dm₁,₃→₁ // dm₄,₅→₄ // dm₄,₂→₄ // dm₄,₆→₄ // ExpandAB // DUForm
```

DU₁[] DU₄[]

Rotate a crossing:

```
$TD = 3;
(cup₁ dS₅[R₂,₅] cap₃ cup₄ cap₆ // dm₁,₂→₁ // dm₁,₃→₁ // dm₄,₅->₄ // dm₄,₆→₄)  - dS₄[R₁,₄] // ExpandAB
```

0

Reidemeister 1 curls, the two positives agree with the two negatives and they are inverses and agree with v, v^{-1}.

```
$TD = 2; {
   (R₁,₂ cap₃ // dm₁,₃→₁ // dm₁,₂→₁) - (R₁,₂ cup₃ // dm₂,₃→₂ // dm₂,₁→₁),
   (dS₂[R₁,₂] cup₃ // dm₁,₃→₁ // dm₁,₂→₁) - (dS₂[R₁,₂] cap₃ // dm₂,₃→₂ // dm₂,₁→₁),
   (dS₂[R₁,₂] cup₃ // dm₁,₃→₁ // dm₁,₂→₁) ** (R₁,₂ cap₃ // dm₁,₃→₁ // dm₁,₂→₁),
   (dS₂[R₁,₂] cup₃ // dm₁,₃→₁ // dm₁,₂→₁) - v₁} // ExpandAB // DUForm
```

{0, 0, DU₁[], 0}

The Central Element

```
c₁ = α U₋₁[b] U₁[] + β U₋₁[] U₁[a];
UB[c₁, #] & /@ {U₋₁[] U₁[a], U₋₁[] U₁[x], U₋₁[y] U₁[], U₋₁[b] U₁[]}
```

{0, 0, 0, 0}

Commuting Exponentials

Commuting e^a with e^x:

```
$TD = 5; 𝕆[U₁[x → x, a → a], e^(ħ (μ x+ν a))] == 𝕆[U₁[a → a, x → x], e^(ħ (μ e^(-ħ α ν) x+ν a))]
```

True

Commuting e^b with e^y:

```
$TD = 5; 𝕆[U₁[b → b, y → y], 𝕖^(ℏ (μ y+ν b))] == 𝕆[U₁[y → y, b → b], 𝕖^(ℏ (μ 𝕖^(-ℏ β ν) y+ν b))]
```

```
True
```

The co-product of e^a:

```
$TD = 5; Δ_{1→1,2}[𝕆[U₁[a → a], 𝕖^(ℏ μ a)]] == 𝕆[U₁[a₁ → a] U₂[a₂ → a], 𝕖^(ℏ μ a₁) 𝕖^(ℏ μ a₂)]
```

```
True
```

The co-product of e_q^x:

```
$TD = 5 ; (𝕆[U₁[x → x], 𝕖_q[ℏ μ x]] // Δ_{1→1,2} // ExpandAB) ==
 𝕆[U₁[a₁ → a, x₁ → x] U₂[x₂ → x], 𝕖_q[ℏ μ 𝕖^(-ℏ β a₁) x₂] 𝕖_q[ℏ μ x₁]]
```

```
True
```

The triple co-product of e_q^x:

```
$TD = 5; (𝕆[U₁[x → x], 𝕖_q[ℏ μ x]] // Δ_{1→1,2,3} // ExpandAB) -
 𝕆[U₁[a₁ → a, x₁ → x] U₂[a₂ → a, x₂ → x] U₃[x₃ → x],
  𝕖_q[ℏ μ 𝕖^(-ℏ β (a₁+a₂)) x₃] 𝕖_q[ℏ μ 𝕖^(-ℏ β a₁) x₂] 𝕖_q[ℏ μ x₁]]
```

```
0
```

The co-product of e_q^y:

```
$TD = 5 ; (𝕆[U₁[y → y], 𝕖_q[ℏ ν y]] // Δ_{1→1,2} // ExpandAB) ==
 𝕆[U₁[y₁ → y] U₂[y₂ → y, b₂ → b], 𝕖_q[ℏ ν y₂] 𝕖_q[ℏ ν 𝕖^(-ℏ α b₂) y₁]]
```

```
True
```

The triple co-product of e_q^y:

```
$TD = 5 ; (𝕆[U₁[y → y], 𝕖_q[ℏ ν y]] // Δ_{1→1,2,3} // ExpandAB) ==
 𝕆[U₁[y₁ → y] U₂[y₂ → y, b₂ → b] U₃[y₃ → y, b₃ → b],
  𝕖_q[ℏ ν y₃] 𝕖_q[ℏ ν 𝕖^(-ℏ α b₃) y₂] 𝕖_q[ℏ ν 𝕖^(-ℏ α (b₂+b₃)) y₁]]
```

```
True
```

The inverse-antipode on e_q^x:

```
$TD = 8;
(𝕆[U₁[x → x], 𝕖_q[ℏ μ x]] // Si₁ // ExpandAB) -
 (𝕆[U₁[a → a, x → x], Sum[((-ℏ μ)^k q^(-k (k+1)/2) 𝕖^(ℏ β k a) x^k)/QFactorial[k, q], {k, 0, $TD}]])
```

```
0
```

Pairing e_q^x with e_q^y:

```
$TD = 5;
$TD *= 2;
lhs = 𝕆[U₋₁[y → y], 𝕖_q[ℏ ν y]] 𝕆[U₁[x → x], 𝕖_q[ℏ μ x]] // Expand // P_{1,-1} // ExpandAB;
($TD /= 2; 𝕖_q[ℏ μ ν] - lhs // ExpandAB)
```

```
0
```

Computing $e^{\mu x} y e^{-\mu x}$:

```
ϕ₀[x_] := 𝕖^x;
ϕ_k_[x_] := Module[{t}, Simplify[1/t^k (𝕖^t - Normal@Series[𝕖^t, {t, 0, k - 1}]) /. t → x]];
```

```
Table[ϕ_k[x], {k, 0, 3}]
```

$$\left\{e^x, \frac{-1+e^x}{x}, \frac{-1+e^x-x}{x^2}, -\frac{2-2e^x+2x+x^2}{2x^3}\right\}$$

```
$TD = 5; {lhs = 𝕆[U₋₁[] U₁[x → x], 𝕖^(ℏ μ x)] ** DU₁[y] ** 𝕆[U₋₁[] U₁[x → x], 𝕖^(-ℏ μ x)],
   rhs = DU₁[y] + μ DU₁[B[1]] ** 𝕆[U₋₁[] U₁[x → x], ϕ₁[ℏ μ (q - 1) x]] -
      μ DU₁[A[1]] ** 𝕆[U₋₁[] U₁[x → x], ϕ₁[ℏ μ (q^-1 - 1) x]],
   lhs - rhs} // Last // ExpandAB // DUForm
```

```
0
```

lhs

$U_{-1}[y]\,U_1[\,] + \mu\,U_{-1}[B[1]]\,U_1[\,] - \frac{1}{2}\mu^2\hbar\,U_{-1}[B[1]]\,U_1[x] +$
$\frac{1}{2}e^{\alpha\beta\hbar}\mu^2\hbar\,U_{-1}[B[1]]\,U_1[x] - \mu\,U_{-1}[\,]\,U_1[A[1]] + \frac{1}{6}\mu^3\hbar^2\,U_{-1}[B[1]]\,U_1[x,x] -$
$\frac{1}{3}e^{\alpha\beta\hbar}\mu^3\hbar^2\,U_{-1}[B[1]]\,U_1[x,x] + \frac{1}{6}e^{2\alpha\beta\hbar}\mu^3\hbar^2\,U_{-1}[B[1]]\,U_1[x,x] -$
$\frac{1}{2}\mu^2\hbar\,U_{-1}[\,]\,U_1[A[1],x] + \frac{1}{2}e^{\alpha\beta\hbar}\mu^2\hbar\,U_{-1}[\,]\,U_1[A[1],x] - \frac{1}{24}\mu^4\hbar^3\,U_{-1}[B[1]]\,U_1[x,x,x] +$
$\frac{1}{8}e^{\alpha\beta\hbar}\mu^4\hbar^3\,U_{-1}[B[1]]\,U_1[x,x,x] - \frac{1}{8}e^{2\alpha\beta\hbar}\mu^4\hbar^3\,U_{-1}[B[1]]\,U_1[x,x,x] +$
$\frac{1}{24}e^{3\alpha\beta\hbar}\mu^4\hbar^3\,U_{-1}[B[1]]\,U_1[x,x,x] - \frac{1}{6}\mu^3\hbar^2\,U_{-1}[\,]\,U_1[A[1],x,x] +$
$\frac{1}{3}e^{\alpha\beta\hbar}\mu^3\hbar^2\,U_{-1}[\,]\,U_1[A[1],x,x] - \frac{1}{6}e^{2\alpha\beta\hbar}\mu^3\hbar^2\,U_{-1}[\,]\,U_1[A[1],x,x] +$
$\frac{1}{120}\mu^5\hbar^4\,U_{-1}[B[1]]\,U_1[x,x,x,x] - \frac{1}{30}e^{\alpha\beta\hbar}\mu^5\hbar^4\,U_{-1}[B[1]]\,U_1[x,x,x,x] +$
$\frac{1}{20}e^{2\alpha\beta\hbar}\mu^5\hbar^4\,U_{-1}[B[1]]\,U_1[x,x,x,x] - \frac{1}{30}e^{3\alpha\beta\hbar}\mu^5\hbar^4\,U_{-1}[B[1]]\,U_1[x,x,x,x] +$
$\frac{1}{120}e^{4\alpha\beta\hbar}\mu^5\hbar^4\,U_{-1}[B[1]]\,U_1[x,x,x,x] - \frac{1}{24}\mu^4\hbar^3\,U_{-1}[\,]\,U_1[A[1],x,x,x] +$
$\frac{1}{8}e^{\alpha\beta\hbar}\mu^4\hbar^3\,U_{-1}[\,]\,U_1[A[1],x,x,x] - \frac{1}{8}e^{2\alpha\beta\hbar}\mu^4\hbar^3\,U_{-1}[\,]\,U_1[A[1],x,x,x] +$
$\frac{1}{24}e^{3\alpha\beta\hbar}\mu^4\hbar^3\,U_{-1}[\,]\,U_1[A[1],x,x,x] - \frac{1}{120}\mu^5\hbar^4\,U_{-1}[\,]\,U_1[A[1],x,x,x,x] +$
$\frac{1}{30}e^{\alpha\beta\hbar}\mu^5\hbar^4\,U_{-1}[\,]\,U_1[A[1],x,x,x,x] - \frac{1}{20}e^{2\alpha\beta\hbar}\mu^5\hbar^4\,U_{-1}[\,]\,U_1[A[1],x,x,x,x] +$
$\frac{1}{30}e^{3\alpha\beta\hbar}\mu^5\hbar^4\,U_{-1}[\,]\,U_1[A[1],x,x,x,x] - \frac{1}{120}e^{4\alpha\beta\hbar}\mu^5\hbar^4\,U_{-1}[\,]\,U_1[A[1],x,x,x,x]$

$TD = 3; $\mathbb{O}\left[U_{-1}[y\to y]\,U_1[\,],\ e^{-\hbar\nu y}\right] ** \mathbb{O}\left[U_{-1}[\,]\,U_1[x\to x],\ e^{\hbar\mu x}\right] **$
$\mathbb{O}\left[U_{-1}[y\to y]\,U_1[\,],\ e^{\hbar\nu y}\right] ** \mathbb{O}\left[U_{-1}[\,]\,U_1[x\to x],\ e^{-\hbar\mu x}\right]$ // ExpandAB // DUForm

$DU_1[\,] + \beta\mu\nu\hbar^2\,DU_1[a] - \alpha\mu\nu\hbar^2\,DU_1[b] + \alpha\beta\mu^2\nu\hbar^3\,DU_1[x] +$
$\alpha\beta\mu\nu^2\hbar^3\,DU_1[y] - \frac{1}{2}\beta^2\mu\nu\hbar^3\,DU_1[a,a] + \frac{1}{2}\alpha^2\mu\nu\hbar^3\,DU_1[b,b]$